

Key to Practical 8

Space Invaders (Part 1)

Step 1

```

FillScreen      ; Save registers on the stack.
                 movem.l d7/a0,-(a7)

                 ; A0 points to the video memory.
                 ; This instruction is identical to : movea.l #VIDEO_START,a0
                 lea     VIDEO_START,a0

                 ; Initialize the loop counter (D7.W).
                 ; The size of the copy is 32 bits wide, i.e. 4 bytes wide.
                 ; So, the number of iterations is the size in bytes divided by 4.
                 ; DBRA will be used for the loop,
                 ; So, D7.W must hold the number of iterations minus 1 (see Chapter 1).
                 move.w  #VIDEO_SIZE/4-1,d7

\loop           ; Copy the data into the video memory
                 ; and increment the address.
                 move.l  d0,(a0)+
                 dbra    d7,\loop

                 ; Restore registers from the stack and return from subroutine.
                 movem.l (a7)+,d7/a0
                 rts

```

Step 2

```

HLines      ; Save registers on the stack.
            movem.l d6/d7/a0,-(a7)

            ; A0 points to the video memory.
            lea     VIDEO_START,a0

            ; D7.W = Loop counter
            ;       = Number of iterations - 1 (because DBRA is used).
            ; -----
            ; Number of iterations = Number of white and black stripes
            ; Height of a white stripe = 8 pixels
            ; Height of a black stripe = 8 pixels
            ; Number of white and black stripes = Height of the window / 2x8
            move.l  #VIDEO_HEIGHT/16-1,d7

\loop       ; Draw a white stripe (8 white lines).
            ; -----
            ; D6.W = Loop counter
            ;       = Number of iterations - 1 (because DBRA is used).
            ; -----
            ; Number of iterations = Number of long words
            ; Number of long words = Number of bytes / 4
            ; Number of bytes = BYTE_PER_LINE x Height of a white stripe
            ; Height of a stripe = 8 pixels
            move.w  #BYTE_PER_LINE*8/4-1,d6
\white_loop move.l  #$ffffffff,(a0)+
            dbra   d6,\white_loop

            ; Draw a black stripe (8 black lines).
            move.w  #BYTE_PER_LINE*8/4-1,d6
\black_loop clr.l   (a0)+
            dbra   d6,\black_loop

            ; Branch to loop as long as there are
            ; white and black stripes to draw.
            dbra   d7,\loop

            ; Restore registers from the stack and return from subroutine.
            movem.l (a7)+,d6/d7/a0
            rts

```